6 Loops

"Repetition is the reality and the seriousness of life."
—Soren Kierkegaard

"What's the key to comedy? Repetition. What's the key to comedy? Repetition."
—Anonymous

In this chapter:
- The concept of iteration.
- Two types of loops: "while," and "for." When do we use them?
- Iteration in the context of computer graphics.

6.1 What is iteration? I mean, what is iteration? Seriously, what is iteration?

Iteration is the generative process of repeating a set of rules or steps over and over again. It is a fundamental concept in computer programming and we will soon come to discover that it makes our lives as coders quite delightful. Let's begin.

For the moment, think about legs. Lots and lots of legs on our little Zoog. If we had only read Chapter 1 of this book, we would probably write some code as in Example 6-1.

Example 6-1: Many lines

```
size(200,200);
background(255);

// Legs
stroke(0);
line(50,60,50,80);
line(60,60,60,80);
line(70,60,70,80);
line(80,60,80,80);
line(90,60,90,80);
line(100,60,100,80);
line(110,60,110,80);
line(120,60,120,80);
line(130,60,130,80);
line(140,60,140,80);
line(150,60,150,80);
```



fig. 6.1

In the above example, legs are drawn from x = 50 pixels all the way to x = 150 pixels, with one leg every 10 pixels. Sure, the code accomplishes this, however, having learned variables in Chapter 4, we can make some substantial improvements and eliminate the hard-coded values.

First, we set up variables for each parameter of our system: the legs' x, y locations, length, and the spacing between the legs. Note that for each leg drawn, only the x value changes. All other variables stay the same (but they could change if we wanted them to!).

Example 6-2: Many lines with variables

```
size(200,200);
background(0);

// Legs
stroke(255);

int y = 80;          // Vertical location of each line
int x = 50;          // Initial horizontal location for first line
int spacing = 10;    // How far apart is each line
int len = 20;        // Length of each line

line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);

x = x + spacing;
line(x,y,x,y+len);
```

Draw the first leg.

Add spacing so the next leg appears 10 pixels to the right.

Continue this process for each leg, repeating it over and over.

Not too bad, I suppose. Strangely enough, although this is technically more efficient (we could adjust the spacing variable, for example, by changing only one line of code), we have taken a step backward, having produced twice as much code! And what if we wanted to draw 100 legs? For every leg, we need two lines of code. That's 200 lines of code for 100 legs! To avoid this dire, carpal-tunnel inducing problem, we want to be able to say something like:

Draw one line one hundred times.

Aha, only one line of code!

Obviously, we are not the first programmers to reach this dilemma and it is easily solved with the very commonly used *control structure*—the *loop*. A loop structure is similar in syntax to a conditional

(see Chapter 5). However, instead of asking a yes or no question to determine whether a block of code should be executed one time, our code will ask a yes or no question to determine *how many times* the block of code should be *repeated*. This is known as iteration.

6.2 "WHILE" Loop, the Only Loop You Really Need

There are three types of loops, the ***while*** loop, the ***do-while*** loop, and the ***for*** loop. To get started, we are going to focus on the ***while*** loop for a little while (sorry, couldn't resist). For one thing, the only loop you really need is ***while***. The ***for*** loop, as we will see, is simply a convenient alternative, a great shorthand for simple counting operations. ***Do-while***, however, is rarely used (not one example in this book requires it) and so we will ignore it.

Just as with conditional (***if/else***) structures, a ***while*** loop employs a boolean test condition. If the test evaluates to true, the instructions enclosed in curly brackets are executed; if it is false, we continue on to the next line of code. The difference here is that the instructions inside the ***while*** block continue to be executed over and over again until the test condition becomes false. See Figure 6.2.



fig. 6.2

Let's take the code from the legs problem. Assuming the following variables . . .

```
int y = 80;          // Vertical location of each line
int x = 50;          // Initial horizontal location for first line
int spacing = 10;    // How far apart is each line
int len = 20;        // Length of each line
```

. . . we had to manually repeat the following code:

```
stroke(255);
line(x,y,x,y+len);      // Draw the first leg

x = x + spacing;        // Add "spacing" to x
line(x,y,x,y+len);      // The next leg is 10 pixels to the right


x = x + spacing;        // Add "spacing" to x
line(x,y,x,y+len);      // The next leg is 10 pixels to the right


x = x + spacing;        // Add ''spacing" to x
line(x,y,x,y+len);      // The next leg is 10 pixels to the right

// etc. etc. repeating with new legs
```

Now, with the knowledge of the existence of ***while*** loops, we can rewrite the code as in Example 6-3, adding a variable that tells us when to stop looping, that is, at what pixel the legs stop.

Example 6-3: While loop

```
int endLegs = 150;

stroke(0);
while (x <= endLegs) {
  line(x,y,x,y+len);
  x = x + spacing;
}
```

A variable to mark where the legs end.

Draw each leg inside a ***while*** loop.



fig. 6.3

Instead of writing "*line(x,y,x,y+len);*" many times as we did at first, we now write it only *once inside of the* ***while*** *loop*, saying "as long as *x* is less than 150, draw a line at *x*, all the while incrementing *x*." And so what took 21 lines of code before, now only takes four!

In addition, we can change the spacing variable to generate more legs. The results are shown in Figure 6.4.

```
int spacing = 4;


while (x <= endLegs) {
  line(x,y,x,y+len); // Draw EACH leg
  x = x + spacing;
}
```

A smaller spacing value results in legs closer together.



fig. 6.4

Let's look at one more example, this time using rectangles instead of lines, as shown in Figure 6.5, and ask three key questions.



fig. 6.5

1. What is the initial condition for your loop? Here, since the first rectangle is at *y* location 10, we want to start our loop with $y = 10$.

   ```
   int y = 10;
   ```

2. When should your loop stop? Since we want to display rectangles all the way to the bottom of the window, the loop should stop when *y* is greater than height. In other words, we want the loop to keep going *as long as y is less than height*.

```
while (y < 100) {
  // Loop!
}
```

3. What is your loop operation? In this case, each time through the loop, we want to draw a new rectangle below the previous one. We can accomplish this by calling the ***rect()*** function and incrementing *y* by 20.

```
rect(100,y,100,10);
y = y + 20;
```

Putting it all together:

```
int y = 10;

while (y < height) {
  rect(100,y,100,10);


  y = y + 20;
}
```

Initial condition.

The loop continues while the boolean expression is true. Therefore, the loop stops when the boolean expression is false.

We increment *y* each time through the loop, drawing rectangle after rectangle until *y* is no longer less than height.

Exercise 6-1: Fill in the blanks in the code to recreate the following screenshots.



```
size(200,200);
background(255);
  int y = 0;
  while (________) {
    stroke(0);
    line(_______,_______,_______,_______);
    y = ________;
  }
```



```
size(200,200);
background(255);
  float w = ________;
  while (________) {
    stroke(0);
    fill(________);
    ellipse(_______,_______,_______,_______);
    __________20;
  }
```

6.3 "Exit" Conditions

Loops, as you are probably starting to realize, are quite handy. Nevertheless, there is a dark, seedy underbelly in the world of loops, where nasty things known as *infinite loops* live. See Figure 6.6.



fig. 6.6

Examining the "legs" in Example 6-3, we can see that as soon as *x* is greater than 150, the loop stops. And this always happens because *x* increments by "spacing", which is always a positive number. This is not an accident; whenever we embark on programming with a loop structure, we must make sure that the exit condition for the loop will eventually be met!

Processing will not give you an error should your exit condition never occur. The result is Sisyphean, as your loop rolls the boulder up the hill over and over and over again to infinity.

Example 6-4: Infinite loop. Don't do this!

```
int x = 0;
while (x < 10){
  println(x);
  x = x - 1;
}
```

Decrementing *x* results in an infinite loop here because the value of *x* will never be 10 or greater. Be careful!

For kicks, try running the above code (make sure you have saved all your work and are not running some other mission-critical software on your computer). You will quickly see that *Processing* hangs. The only way out of this predicament is probably to force-quit *Processing*. Infinite loops are not often as obvious as in Example 6-4. Here is another flawed program that will *sometimes* result in an infinite loop crash.

Example 6-5: Another infinite loop. Don't do this!

```
int y = 80;                  // Vertical location of each line
int x = 0;                   // Horizontal location of first line
int spacing = 10;            // How far apart is each line
int len = 20;                // Length of each line
int endLegs = 150;           // Where should the lines stop?

void setup() {
  size(200,200);
}

void draw() {
  background(0);
  stroke(255);

  x = 0;
  spacing = mouseX / 2;
```

The spacing variable, which sets the distance in between each line, is assigned a value equal to ***mouseX*** divided by two.

```
  while (x <= endLegs) {
    line(x,y,x,y+len);

    x = x + spacing;
  }
}
```

Exit Condition — when *x* is greater than ***endlegs.***

Incrementation of *x*. *x* always increases by the value of spacing. What is the range of possible value for spacing?

Will an infinite loop occur? We know we will be stuck looping forever if *x* never is greater than 150. And since *x* increments by spacing, if spacing is zero (or a negative number) *x* will always remain the same value (or go down in value.)

Recalling the ***constrain()*** function described in Chapter 4, we can guarantee no infinite loop by constraining the value of spacing to a positive range of numbers:

```
int spacing = constrain(mouseX/2, 1, 100);
```

Using ***constrain()*** to ensure the exit condition is met.

Since spacing is directly linked with the necessary exit condition, we enforce a specific range of values to make sure no infinite loop is ever reached. In other words, in pseudocode we are saying: *"Draw a series of lines spaced out by N pixels where N can never be less than 1!"*

This is also a useful example because it reveals an interesting fact about ***mouseX***. You might be tempted to try putting ***mouseX*** directly in the incrementation expression as follows:

```
while (x <= endLegs) {
  line(x,y,x,y+len);
  x = x + mouseX/2;
}
```

Placing ***mouseX*** inside the loop is not a solution to the infinite loop problem.

Wouldn't this solve the problem, since even if the loop gets stuck as soon as the user moves the mouse to a horizontal location greater than zero, the exit condition would be met? It is a nice thought, but one that is sadly quite flawed. ***mouseX*** and ***mouseY*** are updated with new values at the beginning of each cycle through ***draw()***. So even if the user moves the mouse to *X* location 50 from location 0, ***mouseX*** will never know this new value because it will be stuck in its infinite loop and not able to get to the next cycle through ***draw()***.

6.4 "FOR" Loop

A certain style of ***while*** loop where one value is incremented repeatedly (demonstrated in Section 6.2) is particularly common. This will become even more evident once we look at arrays in Chapter 9. The ***for*** loop is a nifty shortcut for commonly occurring ***while*** loops. Before we get into the details, let's talk through some common loops you might write in *Processing* and how they are written as a ***for*** loop.

Start at 0 and count up to 9.	`for (int i = 0; i < 10; i = i + 1)`
Start at 0 and count up to 100 by 10.	`for (int i = 0; i < 101; i = i + 10)`
Start at 100 and count down to 0 by 5.	`for (int i = 100; i >= 0; i = i - 5)`

Looking at the above examples, we can see that a ***for*** loop consists of three parts:

- **Initialization**—Here, a variable is declared and initialized for use within the body of the loop. This variable is most often used inside the loop as a counter.
- **Boolean Test**—This is exactly the same as the boolean tests found in conditional statements and ***while*** loops. It can be any expression that evaluates to true or false.
- **Iteration Expression**—The last element is an instruction that you want to happen with each loop cycle. Note that the instruction is executed at the end of each cycle through the loop. (You can have multiple iteration expressions, as well as variable initializations, but for the sake of simplicity we will not worry about this now.)



fig. 6.7

In English, the above code means: repeat this code 10 times. Or to put it even more simply: count from zero to nine!

To the machine, it means the following:

- Declare a variable *i*, and set its initial value to 0.
- While *i* is less than 10, repeat this code.
- At the end of each iteration, add one to *i*.

A ***for*** loop can have its own variable just for the purpose of counting. A variable not declared at the top of the code is called a ***local variable.*** We will explain and define it shortly.

Increment/Decrement Operators

The shortcut for adding or subtracting one from a variable is as follows:

`x++;` is equivalent to: `x = x + 1;` meaning: "increment *x* by 1" or "add 1 to the current value of *x*"

`x--;` is equivalent to: `x = x - 1;`

We also have:
`x+=2;` same as `x=x+2;`
`x*=3;` same as `x=x*3;`
and so on.

The same exact loop can be programmed with the ***while*** format:

```
int i = 0;
while (i < 10) {
  i++;
  //lines of code to execute here
}
```

This is the translation of the ***for*** loop, using a ***while*** loop.

Rewriting the leg drawing code to use a ***for*** statement looks like this:

Example 6-6: Legs with a *for* loop

```
int y = 80;                 // Vertical location of each line
int spacing = 10;           // How far apart is each line
int len = 20;               // Length of each line

for (int x = 50; x <= 150; x += spacing) {
  line(x,y,x,y+len);
}
```

Translation of the legs ***while*** loop to a ***for*** loop.

*Exercise 6-2: Rewrite Exercise 6-1 using a **for** loop.*



```
size(200,200);
background(255);

  for (int y =________;________;________) {
    stroke(0);
    line(________,________,________,________);
  }
```



```
size(200,200);
background(255);
  for (________;________;________-=20) {
    stroke(0);
    fill(________);
    ellipse(________,________,________,
  ________);
    ellipse(________,________,________,
  ________);
  }
```

Exercise 6-3: Following are some additional examples of loops. Match the appropriate screenshot with the loop structure. Each example assumes the same four lines of initial code.




```
size(300,300);       // Just setting up the size
background(255);     // Black background
stroke(0);           // Shapes have white lines
noFill();            // Shapes are not filled in
```

_____
```
for (int i = 0; i < 10; i++) {
  rect(i*20,height/2, 5, 5);
}
```

_____
```
int i = 0;
while (i < 10) {
  ellipse(width/2,height/2, i*10, i*20);
  i++;
}
```

_____
```
for (float i = 1.0; i < width; i *= 1.1) {
  rect(0,i,i,i*2);
}
```

_____
```
int x = 0;
for (int c = 255; c > 0; c -= 15) {
  fill(c);
  rect(x,height/2,10,10);
  x = x + 10;
}
```

6.5 Local vs. Global Variables (AKA "Variable Scope")

Up until this moment, any time that we have used a variable, we have declared it at the top of our program above ***setup()***.

```
int x = 0;
void setup() {

}
```

We have always declared our variables at the top of our code.

This was a nice simplification and allowed us to focus on the fundamentals of declaring, initializing, and using variables. Variables, however, can be declared anywhere within a program and we will now look at what it means to declare a variable somewhere other than the top and how we might go about choosing the right location for declaring a variable.

Imagine, for a moment, that a computer program is running your life. And in this life, variables are pieces of data written on post-its that you need to remember. One post-it might have the address of a restaurant for lunch. You write it down in the morning and throw it away after enjoying a nice turkey burger. But another post-it might contain crucial information (such as a bank account number), and you save it in a safe place for years on end. This is the concept of *scope*. Some variables exist (i.e., are accessible) throughout the entire course of a program's life—*global variables*—and some live temporarily, only for the brief moment when their value is required for an instruction or calculation—*local variables*.

In *Processing*, global variables are declared at the top of the program, outside of both ***setup()*** and ***draw()***. These variables can be used in any line of code anywhere in the program. This is the easiest way to use a variable since you do not have to remember when you can and cannot use that variable. You can *always* use that variable (and this is why we started with global variables only).

Local variables are variables declared within a block of code. So far, we have seen many different examples of blocks of code: ***setup()***, ***draw()***, ***mousePressed()***, and ***keyPressed()***, ***if*** statements, and ***while*** and ***for*** loops.

A local variable declared within a block of code is only available for use inside that specific block of code where it was declared. If you try to access a local variable outside of the block where it was declared, you will get this error:

"***No accessible field named*** "variableName" ***was found***"

This is the same exact error you would get if you did not bother to declare the variable "variableName" at all. *Processing* does not know what it is because no variable with that name exists within the block of code you happen to be in.

Here is an example where a local variable is used inside of ***draw()*** for the purpose of executing a ***while*** loop.

Example 6-7: Local variable

```
void setup() {
  size(200,200);
}
void draw() {
  background(0);
```

X is not available! It is local to the ***draw()*** block of code.

```
int x = 0;
  while (x < width) {
    stroke(255);
    line(x,0,x,height);
    x += 5;
  }
}
```

X is available! Since it is declared within the ***draw()*** block of code, it is available here. Notice, however, that it is not available inside ***draw()*** above where it is declared. Also, it is available inside the ***while*** block of code because ***while*** is inside of ***draw ().***

```
void mousePressed() {
  println("The mouse was pressed!");
}
```

X is not available! It is local to the ***draw()*** block of code.

Why bother? Couldn't we just have declared *x* as a global variable? While this is true, since we are only using *x* within the ***draw()*** function, it is wasteful to have it as a global variable. It is more efficient and ultimately less confusing when programming to declare variables only within the scope of where they are necessary. Certainly, many variables *need* to be global, but this is not the case here.

A ***for*** loop offers up a spot for a local variable within the "initialization" part:

```
for (int i = 0; i < 100; i+=10) {
  stroke(255);
  fill(i);
  rect(i,0,10,height);
}
```

i is only available inside the ***for*** loop.

It is not required to use a local variable in the ***for*** loop, however, it is usually convenient to do so.

It is theoretically possible to declare a local variable with the same name as a global variable. In this case, the program will use the local variable within the current scope and the global variable outside of that scope. As a general rule, it is better to never declare multiple variables with the same name in order to avoid this type of confusion.

Exercise 6–4: Predict the results of the following two programs. Test your theory by running them.




```
//SKETCH #1: Global
"count"
int count = 0;

void setup() {
  size(200,200);
}

void draw() {
  count = count + 1;
  background(count);
}
```

_____

```
//SKETCH #2: Local
"count"
void setup() {
  size(200,200);
}

void draw() {
  int count = 0;
  count = count = 1;
  background(count);
}
```

_____

6.6 Loop Inside the Main Loop

The distinction between local and global variables moves us one step further toward successfully integrating a loop structure into Zoog. Before we finish this chapter, I want to take a look at one of the most common points of confusion that comes with writing your first loop in the context of a "dynamic" *Processing* sketch.

Consider the following loop (which happens to be the answer to Exercise 6-2). The outcome of the loop is shown in Figure 6.8.



fig. 6.8

```
for (int y = 0; y < height; y+=10) {
  stroke(0);
  line(0,y,width,y);
}
```

Let's say we want to take the above loop and display each line one at a time so that we see the lines appear animated from top to bottom. Our first thought might be to take the above loop and bring it into a dynamic *Processing* sketch with ***setup()*** and ***draw()***.

```
void setup() {
  size(200,200);
}

void draw() {
  background(255);
  for (int y = 0; y < height; y+=10) {
    stroke(0);
    line(0,y,width,y);
  }
}
```

If we read the code, it seems to make sense that we would see each line appear one at a time. "Set up a window of size 200 by 200 pixels. Draw a black background. Draw a line at *y* equals 0. Draw a line at *y* equals 10. Draw a line at *y* equals 20."

Referring back to Chapter 2, however, we recall that *Processing* does not actually update the display window until the end of ***draw()*** is reached. This is crucial to remember when using ***while*** and ***for*** loops. These loops serve the purpose of repeating something in the context of *one cycle* through ***draw()***. They are a loop inside of the sketch's main loop, ***draw()***.

Displaying the lines one at a time is something we can do with a global variable in combination with the very looping nature of ***draw()*** itself

Example 6-8: Lines one at a time

```
int y = 0;

void setup() {
  size(200,200);
  background(0);
  frameRate(5);
}
```

No ***for*** loop here. Instead, a global variable.

Slowing down the frame rate so we can easily see the effect.

```
void draw() {
  // Draw a line
  stroke(255);
  line(0,y,width,y);
  // Increment y
  y += 10;
}
```

Only one line is drawn each time through ***draw()***.

The logic of this sketch is identical to Example 4-3, our first motion sketch with variables. Instead of moving a circle across the window horizontally, we are moving a line vertically (but not clearing the background for each frame).

*Exercise 6-5: It is possible to achieve the effect of rendering one line at a time using a **for** loop. See if you can figure out how this is done. Part of the code is below.*

```
int endY;

void setup() {
  size(200,200);
  frameRate(5);
  endY = ________;
}

void draw() {
  background(0);
  for (int y = ________; ________; ________) {
    stroke(255);
    line(0,y,width,y);
  }
  ________;
}
```

Using a loop inside ***draw()*** also opens up the possibility of interactivity. Example 6-9 displays a series of rectangles (from left to right), each one colored with a brightness according to its distance from the mouse.

Example 6-9: Simple while loop with interactivity

```
void setup() {
  size(255,255);
  background(0);
}

void draw() {
  background(0);
  // Start with i as 0
  int i = 0;
  // While i is less than the width of the window
  while (i < width) {
    noStroke();
```



fig. 6.9

```
    float distance = abs(mouseX - i);

    fill(distance);
    rect(i,0,10,height);

    // Increase i by 10
    i += 10;
  }
}
```

The distance between the current rectangle and the mouse is equal to the absolute value of the difference between *i* and ***mouseX***.

That distance is used to fill the color of a rectangle at horizontal location *i*.

*Exercise 6-6: Rewrite Example 6-9 using a **for** loop.*

6.7 Zoog grows arms.

We last left Zoog bouncing back and forth in our *Processing* window. This new version of Zoog comes with one small change. Example 6-10 uses a ***for*** loop to add a series of lines to Zoog's body, resembling arms.

Example 6-10: Zoog with arms

```
int x = 100;
int y = 100;
int w = 60;
int h = 60;
int eyeSize = 16;
int speed = 1;

void setup() {

  size(200,200);
  smooth();
}

void draw(){

  // Change the x location of Zoog by speed
  x = x + speed;

  // If we've reached an edge, reverse speed (i.e. multiply it by -1)
  //(Note if speed is a + number, square moves to the right,- to the left)
  if ((x > width)||(x < 0)) {
    speed = speed * -1;
  }

  background(255); // Draw a white background

  // Set ellipses and rects to CENTER mode
  ellipseMode(CENTER);
  rectMode(CENTER);

  // Draw Zoog's arms with a for loop
  for (int i = y+5; i < y + h; i+=10) {
    stroke(0);
    line(x-w/3,i,x+w/3,i);
  }
```



fig. 6.10

Arms are incorporated into Zoog's design with a ***for*** loop drawing a series if lines.

```
  // Draw Zoog's body
  stroke(0);
  fill(175);
  rect(x,y,w/6,h*2);

  // Draw Zoog's head
  fill(255);
  ellipse(x,y-h/2,w,h);

  // Draw Zoog's eyes
  fill(0);
  ellipse(x-w/3,y-h/2,eyeSize,eyeSize*2);
  ellipse(x+w/3,y-h/2,eyeSize,eyeSize*2);

  // Draw Zoog's legs
  stroke(0);
  line(x-w/12,y+h,x-w/4,y+h+10);
  line(x+w/12,y+h,x+w/4,y+h+10);
}
```

We can also use a loop to draw multiple instances of Zoog by placing the code for Zoog's body inside of a ***for*** loop. See Example 6–11.

Example 6-11: Multiple Zoogs

```
int w = 60;
int h = 60;
int eyeSize = 16;

void setup() {
  size(400,200);
  smooth();
}

void draw() {
background(255);
ellipseMode(CENTER);
rectMode(CENTER);

int y = height/2;

// Multiple versions of Zoog
for (int x = 80; x < width; x+= 80) {
  // Draw Zoog's body
  stroke(0);
  fill(175);
  rect(x,y,w/6,h*2);

  // Draw Zoog's head
  fill(255);
  ellipse(x,y-h/2,w,h);

  // Draw Zoog's eyes
  fill(0);
  ellipse(x-w/3,y-h/2,eyeSize,eyeSize*2);
  ellipse(x+w/3,y-h/2,eyeSize,eyeSize*2);
```



fig. 6.11

The variable x is now included in a ***for*** loop, in order to iterate and display multiple Zoogs!

```
  // Draw Zoog's legs
  stroke(0);
  line(x-w/12,y+h,x-w/4,y+h+10);
  line(x+w/12,y+h,x+w/4,y+h+10); }
}
```

Exercise 6-7: Add something to your design using a ***for*** *or* ***while*** *loop. Is there anything you already have which could be made more efficient with a loop?*

Exercise 6-8: Create a grid of squares (each colored randomly) using a ***for*** *loop. (Hint: You will need two* ***for*** *loops!) Recode the same pattern using a "while" loop instead of "for."*